Instructor:

BRIJ MOHAN
Hansraj College,
University of Delhi, Delhi.

Maple 2025 provides innovative new tools for comparing expressions.

Expression Tools for Comparing Expressions

- Let us compare two large expressions but we couldn't easily see how they differed? For example:

```
> e1 := 1/4*(2*T[x,j,g]*T[i,c,k]-2*T[x,j,c]*T[i,g,k]-2*d[j](T[x,c,i],[X])+2*d[x](T[c,j,i],[X]))*g_[a,b]+1/4*(T[a,x,b]+T[b,x,a])*T[c,j,i]-1/4*T[b,x,j]*T[c,a,i]-1/2*T[x,j,a]*T[i,b,c]+1/2*T[x,a,b]*T[j,c,i]-1/4*T[a,x,j]*T[c,b,i]+1/4*d[x](T[a,b,c],[X])+1/4*d[x](T[b,a,c],[X])-1/2*d[x](T[c,a,b],[X])-1/4*d[a](T[b,x,c],[X])+1/2*d[b](T[x,a,c],[X])-1/4*d[b](T[a,x,c],[X]);
```

$$e1 := \frac{\left(2\, T_{x,j,g}\, T_{i,c,k} - 2\, T_{x,j,c}\, T_{i,g,k} - 2\, d_j\left(T_{x,c,i}, [X]\right) + 2\, d_x\left(T_{c,j,i}, [X]\right)\right) g_{-a,b}}{4} + \frac{\left(T_{a,x,b} + T_{b,x,a}\right) T_{c,j,i}}{4} - \frac{T_{b,x,j}\, T_{c,a,i}}{4} - \frac{T_{x,j,a}\, T_{i,b,c}}{2} + \frac{T_{x,a,b}\, T_{j,c,i}}{2} - \frac{T_{a,x,j}\, T_{c,b,i}}{4} + \frac{d_x\left(T_{a,b,c}, [X]\right)}{4} + \frac{d_x\left(T_{b,a,c}, [X]\right)}{4} - \frac{d_x\left(T_{c,a,b}, [X]\right)}{2} - \frac{d_a\left(T_{b,x,c}, [X]\right)}{4} + \frac{d_b\left(T_{x,a,c}, [X]\right)}{2} - \frac{d_b\left(T_{a,x,c}, [X]\right)}{4} \tag{1.1}$$

```
> e2 := 1/4*(2*T[x,j,g]*T[i,c,k]-2*T[x,j,c]*T[i,g,k]+2*d[j](T[i,x,c],[X])-2*d[x](T[j,c,i],[X]))*g_[a,b]+1/4*(T[a,x,b]+T[b,x,a])*T[c,j,i]-1/4*T[b,x,j]*T[c,a,i]-1/2*T[x,j,a]*T[i,b,c]+1/2*T[x,a,b]*T[j,c,i]-1/4*T[a,x,j]*T[c,b,i]+1/4*d[x](T[a,b,c],[X])+1/4*d[x](T[b,a,c],[X])-1/2*d[x](T[c,a,b],[X])-1/4*d[a](T[b,x,c],[X])+1/2*d[b](T[x,a,c],[X])-1/4*d[b](T[a,x,c],[X]);
```

$$e2 := \frac{\left(2\, T_{x,j,g}\, T_{i,c,k} - 2\, T_{x,j,c}\, T_{i,g,k} + 2\, d_j\left(T_{i,x,c}, [X]\right) - 2\, d_x\left(T_{j,c,i}, [X]\right)\right) g_{-a,b}}{4} + \frac{\left(T_{a,x,b} + T_{b,x,a}\right) T_{c,j,i}}{4} - \frac{T_{b,x,j}\, T_{c,a,i}}{4} - \frac{T_{x,j,a}\, T_{i,b,c}}{2} + \frac{T_{x,a,b}\, T_{j,c,i}}{2} - \frac{T_{a,x,j}\, T_{c,b,i}}{4} + \frac{d_x\left(T_{a,b,c}, [X]\right)}{4} + \frac{d_x\left(T_{b,a,c}, [X]\right)}{4} - \frac{d_x\left(T_{c,a,b}, [X]\right)}{2} - \frac{d_a\left(T_{b,x,c}, [X]\right)}{4} + \frac{d_b\left(T_{x,a,c}, [X]\right)}{2} - \frac{d_b\left(T_{a,x,c}, [X]\right)}{4} \tag{1.2}$$

```
> evalb(e1=e2);
```

$$\mathit{false} \tag{1.3}$$

```
>
```

- The new package ExpressionTools permits you to perform a visual comparison of two expressions.
- The main command in this package is Compare, which, given two expressions, compares them and highlights their differences:

```
> with(ExpressionTools);
```

$$[Compare, Options] \tag{1.4}$$

```
> Compare(e1, e2);
```

$$\frac{1}{4}\left(\left(2\,T_{x,j,g}\,T_{i,c,k}-2\,T_{x,j,c}\,T_{i,g,k}-2\,d_{\begin{pmatrix} j \\ x \end{pmatrix}}\left(T_{\begin{pmatrix} x \\ j \end{pmatrix},c,i},[X]\right)+2\,d_{\begin{pmatrix} x \\ j \end{pmatrix}}\left(T_{\begin{pmatrix} c \\ i \end{pmatrix},\begin{pmatrix} j \\ x \end{pmatrix},\begin{pmatrix} i \\ c \end{pmatrix}},[X]\right)\right)g_{-a,b}\right)+\frac{\left(T_{a,x,b}+T_{b,x,a}\right)T_{c,j,i}}{4}-\frac{T_{b,x,j}\,T_{c,a,i}}{4}-\frac{T_{x,j,a}\,T_{i,b,c}}{2}+\frac{T_{x,a,b}\,T_{j,c,i}}{2}-\frac{T_{a,x,j}\,T_{c,b,i}}{4}+\frac{d_x\left(T_{a,b,c},[X]\right)}{4}+\frac{d_x\left(T_{b,a,c},[X]\right)}{4}-\frac{d_x\left(T_{c,a,b},[X]\right)}{2}-\frac{d_a\left(T_{b,x,c},[X]\right)}{4}+\frac{d_b\left(T_{x,a,c},[X]\right)}{2}-\frac{d_b\left(T_{a,x,c},[X]\right)}{4} \tag{1.5}$$

In the above example, the two expressions are combined into one. Differing subexpressions are replaced by a vector containing the two corresponding subexpressions, each highlighted in a different color to indicate which of the original expressions it was a part of.

- For smaller expressions (both having length less than 250, an adjustable parameter), the two expressions are printed one after the other, and the differences are simply highlighted by changing their colors:

```
> x1 := (a-b)*(A-B)*c;
```

$$x1 := (a-b)\,(A-B)\,c \tag{1.6}$$

```
> x2 := (b-a)*(B-A)*C;
```

$$x2 := (b-a)\,(B-A)\,C \tag{1.7}$$

```
> Compare(x1, x2);
```

$$(a+-1\,b)\,(A+-1\,B)\,c$$
$$(b+-1\,a)\,(B+-1\,A)\,C \tag{1.8}$$

- If a verification is provided as an option, the comparison ignores any differences it can find that satisfy that verification:

```
> Compare(x1, x2, sign);
```

$$(a-b)\,(A-B)\,c$$
$$(b-a)\,(B-A)\,C \tag{1.9}$$

- For sums, products, and sets, operands are compared according to the best match, even when none of the top-level subexpressions match exactly:

```
> x3, x4 := a/2+b/3+c/4, c/5+b/6+d/7:
> Compare(x3, x4, combine);
```

$$\begin{pmatrix} \frac{a}{2} \\ \frac{d}{7} \end{pmatrix} + \begin{pmatrix} \frac{1}{3} \\ \frac{1}{6} \end{pmatrix} b + \begin{pmatrix} \frac{1}{4} \\ \frac{1}{5} \end{pmatrix} c \quad \textbf{(1.10)}$$

Note in the above that we used the combine option, which forces display in combined format even though the expressions were small.

• Here's a larger example, showing the use of some other options to control the display:

```
> x3 := Vector(2,[w(x), z(x)]) = Vector(2,[1/15*exp(-x)*(-15*c__1+3)+1/15*(10*c__2+2)*exp(4*x)-1/3*exp(x), 1/5*exp(-x)*(-1+5*c__1)+1/5*exp(4*x)*(1+5*c__2)]);
```

$$x3 := \begin{bmatrix} w(x) \\ z(x) \end{bmatrix} = \quad \textbf{(1.11)}$$

$$\begin{bmatrix} \frac{e^{-x}\left(-15\, c_1 + 3\right)}{15} + \frac{\left(10\, c_2 + 2\right) e^{4x}}{15} - \frac{e^{x}}{3} & \ldots \\ \frac{e^{-x}\left(-1 + 5\, c_1\right)}{5} + \frac{e^{4x}\left(1 + 5\, c_2\right)}{5} & \ldots \end{bmatrix}$$

```
> x4 := Vector(2,[w(x), z(x)]) = Vector(2, [1/15*exp(-x)*(-15*_C1+3)+1/15*(10*_C2+2)*exp(4*x)-1/3*exp(x),1/5*exp(-x)*((1+5*_C2)*exp(5*x)-1+5*_C1)]);
```

$$x4 := \begin{bmatrix} w(x) \\ z(x) \end{bmatrix} = \quad \textbf{(1.12)}$$

$$\begin{bmatrix} \frac{e^{-x}\left(-15\, \_C1 + 3\right)}{15} + \frac{\left(10\, \_C2 + 2\right) e^{4x}}{15} - \frac{e^{x}}{3} & \ldots \\ \frac{e^{-x}\left(\left(1 + 5\, \_C2\right) e^{5x} - 1 + 5\, \_C1\right)}{5} & \ldots \end{bmatrix}$$

```
> Compare(x3, x4, Silver, Gold, showvectorbrackets=false, showNULLoperands);
```

$$\begin{bmatrix} w(x) \\ z(x) \end{bmatrix} = \begin{bmatrix} \frac{e^{-x}\left(-15\, c_{1}\, \_C1 + 3\right)}{15} + \cdots \\ \cdots\; e^{-1}\, x^{4}\left(\left(1 + 5 \cdots\right.\right.\; + \cdots \\ \cdots \end{bmatrix} \tag{1.13}$$

• Compare uses the uneval parameter modifier, which means it does not evaluate its arguments before comparing them:

```
> x5 := 'hypergeom'([a,b],[c],x);
```

$$x5 := \mathrm{hypergeom}([a, b], [c], x) \tag{1.14}$$

```
> x6 := 'hypergeom'([b,a],[c],x);
```

$$x6 := \mathrm{hypergeom}([b, a], [c], x) \tag{1.15}$$

```
> evalb(x5=x6);
```

$$\mathit{true} \tag{1.16}$$

```
> Compare(x5, x6);
```

$$\begin{matrix} \mathrm{hypergeom}([a, b], [c], x) \\ \mathrm{hypergeom}([b, a], [c], x) \end{matrix} \tag{1.17}$$

• Note, however, that this could lead to unexpected results:

```
> Compare(x1, subs(C=c, x2), combine, sign);
```

$$\begin{pmatrix} (a-b)(A-B)\,c \\ \mathit{subs}(C=c, x2) \end{pmatrix} \tag{1.18}$$

• So be sure to evaluate results if desired before passing them to Compare:

```
> x3 := subs(C=c, x2);
```

$$x3 := (b - a)\,(B - A)\,c \tag{1.19}$$

```
> Compare(x1, x3, combine, sign);
```

The expressions satisfy the given verification **(1.20)**

Or alternatively you can use the evaluate option to force evaluation:

```
> Compare(x5, x6, evaluate);
```

The expressions are the same **(1.21)**

```
> Compare(x1, subs(C=c, x2), sign, evaluate);
```

The expressions satisfy the given verification **(1.22)**

- A secondary ExpressionTools command called Options allows you to query and set the default values for each option available to the Compare command:

```
> Options();
```

backgroundcolor : [LightGreen, Pink]
combine : default
combinethreshold : 250
evaluate : false
foregroundcolor : [Green, Red]
highlight : default
showNULLoperands : false
showvectorbrackets : true
verification : boolean **(1.23)**

```
> x1 := (2*a-3*b)*(A-B)*c:
> x2 := (3*b-2*a)*(B-A)*C:
> Compare(x1, x2);
```

$$(2\,a + -3\,b)\,(A + -1\,B)\,c$$
$$(3\,b + -2\,a)\,(B + -1\,A)\,C \tag{1.24}$$

```
> Options(combine, highlight, backgroundcolor);
```

combine : default
highlight : default
backgroundcolor : [LightGreen, Pink] **(1.25)**

```
> Options(combine=true, highlight=background, backgroundcolor
  = [Yellow, Orange]);
```

combine : true
highlight : background
backgroundcolor : [Yellow, Orange] **(1.26)**

```
> Compare(x1, x2);
```

$$\left(\begin{pmatrix} 2 \\ -2 \end{pmatrix} a + \begin{pmatrix} -3 \\ 3 \end{pmatrix} b\right)\left((-1)\,A + (-1)\,B\right)\begin{pmatrix} c \\ C \end{pmatrix} \tag{1.27}$$